
/* -*-prolog-*- */

/*
   命題様相論理K, KD45のHilbert流演繹体系での証明を作るPrologプログラム
   SWI-Prologで動作する(他の処理系でも動くかも)
				     by NIDE, N. nide@ics.nara-wu.ac.jp
   Ver. 2014-06-20(第1版)
     初版
   Ver. 2016-01-12(第2版)
     従来、必然オペレータ□を表すのに空リストと同じ記号[]を使っていたが、
     SWI-Prologのver.6以降で[]をオペレータにできなくなったため、その旨の
     説明を追加するとともに、オペレータでなく空リストとして使われている
     []の表記を「[ ]」に変更した(エディタで[]を別な記号に一斉置換するこ
     とで、SWI-Prologのver.6以降での使用にも対応できるようにするため)
   Ver. 2017-09-10(第3版)
     上記と同じ理由で、オペレータとして使われていた「[]」を「#」に変更。
     これにより、エディタでの一斉置換などよる変更を加えることなく、SWI-
     Prologのver.6以降でもそのまま使用できる

   使い方:

	?- deduction(論理式, Tree).
	論理式の証明木を作成してTreeに入れる

	?- prove(論理式).
	論理式の証明木を作成して出力する

	   以上いずれも、デフォルトでは体系Kでの証明を行う
	   もちろんその論理式が証明できないならfalseになる
	   論理式には/\, \/, =>, ~, #, <>の各オペレータが使える(それぞれ
	   ∧∨⊃¬□◇を表す)
	   但し、オペレータが連続する場合は離して書くこと(1つの記号と見な
	   されるのを防ぐため) 例えば<><>pではなく<> <>pと書く
	   従来、必然オペレータ□を表すのに空リストと同じ記号[]を使ってい
	   たが、上述の理由で#に変更されたので注意

	?- set_system_type(k). あるいは ?- set_system_type(kd45).
	体系をKあるいはKD45に切り替える(次に明示的に切り替えるまではその
	体系での証明を行う)

   例:
	?- prove(#(p/\q)=> #p).
	 … Kでの□(p∧q)⊃□pの証明木を出力する …
	true.
	?- prove(#p=> #(p/\q)).
	false.
	?- set_system_type(kd45), prove(#(p/\q)=> # <>p).
	 … KD45での□(p∧q)⊃□◇pの証明木を出力する …
	true.
	?- prove(<> <>p=> <>(p\/q)).
	 … KD45に切り替え済のため、KD45での◇◇p⊃◇(p∨q)の証明木を出力 …
	true.

   但し/\, \/, <>は略記であり、=>, ~, #による正式な形に直されてから
   seq_deduction/2 (シーケントのHilbert流演繹を作り出す述語)に渡される
   従ってseq_deduction述語を直接呼ぶ場合は論理式は=>, ~, #だけで記述
   せねばならない

   prove/1が証明木を出力する場合は以下のフォーマットで出力される

   例:
	?- prove(#(p/\q)=> #p).
	1 [] ax1: p=> ~ ~q=>p
	2 [] ax1: (p=> ~ ~q=>p)=> ~p=>p=> ~ ~q=>p
	3 [1, 2] mp: ~p=>p=> ~ ~q=>p
	 …
	37 [27, 36] mp: ~ (p=> ~q)=>p
	38 [37] nec: # (~ (p=> ~q)=>p)
	39 [] k: # (~ (p=> ~q)=>p)=> # ~ (p=> ~q)=> #p
	40 [38, 39] mp: # ~ (p=> ~q)=> #p
	true.

   一番左はステップ数、その次は前提にした論理式のリスト(よって公理の場合は
   空リスト)、その次は公理や規則の名前(ax1やkは公理、mpは三段論法、necは
   必然化)。その次の「:」の右が論理式で、最後の論理式 # ~(p=> ~q)=> #p は
   proveの引数の略記を正式な形に直したものとなる

   最短の証明が求まるわけではないことに注意。上の例でも途中の ~(p=> ~q)=>p
   を単独で証明させると上の例よりはるかに長いステップ数を要する

   また、特段の高速化もしていない

   内部では、演繹の木をproof_tree(A, Rule, ListOfTree)という再帰的な項で
   表現。第1引数は論理式、第2引数は公理や規則の名前あるいは「hypo」(仮定の
   場合)、第3引数は子ノードのリスト。例えば、仮定pとp=>qからqを導く演繹木は

	proof_tree(q, mp, [
		proof_tree(p, hypo, [ ]),
		proof_tree(p=>q, hypo, [ ])
	])

   のように表す。葉ノード(公理あるいは仮定)は第3引数が空リストとなる。
   葉ノードのうち第2引数を「hypo」とするものが存在しなければ、その木は
   証明の木ということになる
*/

/* オペレータ宣言 */
:-	op(497, yfx, /\), % ∧
	op(498, yfx, \/), % ∨
	op(499, xfy, =>), % ⊃
	op(496, fy, ~),   % ¬
	op(496, fy, #),  % □
	op(496, fy, <>).  % ◇

/* 第1引数に与えた論理式の略記を展開したものを第2引数に入れる */
expand_abbrev(X, X) :-
	atom(X), !.
expand_abbrev(X/\Y, ~(X1=> ~Y1)) :-
	expand_abbrev(X, X1), expand_abbrev(Y, Y1).
expand_abbrev(X\/Y, ~X1=>Y1) :-
	expand_abbrev(X, X1), expand_abbrev(Y, Y1).
expand_abbrev(X=>Y, X1=>Y1) :-
	expand_abbrev(X, X1), expand_abbrev(Y, Y1).
expand_abbrev(~X, ~X1) :-
	expand_abbrev(X, X1).
expand_abbrev(#X, #X1) :-
	expand_abbrev(X, X1).
expand_abbrev(<>X, ~ # ~X1) :-
	expand_abbrev(X, X1).

/*  公理の証明木の作成。第1引数に結論、第2引数にその証明木を入れる */
make_ax1(AX, proof_tree(AX, ax1, [ ])) :-
	AX = A=>_=>A.
make_ax2(AX, proof_tree(AX, ax2, [ ])) :-
	AX = (A=>B=>C)=>(A=>B)=>A=>C.
make_ax3(AX, proof_tree(AX, ax3, [ ])) :-
	AX = (~A=> ~B)=>(~A=> B)=>A.
make_k(AX, proof_tree(AX, k, [ ])) :-
	AX = #(A=>B)=>(#A=> #B).
make_d(AX, proof_tree(AX, d, [ ])) :- system_type(kd45),
	AX = #A=> ~ # ~A.
make_4(AX, proof_tree(AX, 4, [ ])) :- system_type(kd45),
	AX = #A=> # #A.
make_5(AX, proof_tree(AX, 5, [ ])) :- system_type(kd45),
	AX = ~ #A=> # ~ #A.
/* 推論規則の適用。第1引数に結論、第2引数に前提の証明木のリストを与え、
   作った木を第3引数に入れる */
make_mp(B, TreeList, NewTree) :-
	TreeList = [proof_tree(A, _, _), proof_tree(A=>B, _, _)],
	NewTree = proof_tree(B, mp, TreeList).
make_nec(#A, TreeList, NewTree) :-
	TreeList = [proof_tree(A, _, _)],
	NewTree = proof_tree(#A, nec, TreeList).
/* 各補助規則の適用(引数の順序はmake_mpと同じ)、
   あるいは各補助定理の証明木の作成(引数の順序はmake_ax1等と同じ) */
make_weak(A, [Tree1], NewTree) :-
	make_ax1(_, NewTree1),
	make_mp(A, [Tree1, NewTree1], NewTree).
make_mp2(A, [Tree1, Tree2], NewTree) :-
	make_ax2(_, NewTree1),
	make_mp(_, [Tree2, NewTree1], NewTree2),
	make_mp(A, [Tree1, NewTree2], NewTree).
make_mp1(A, [Tree1, Tree2], NewTree) :-
	make_weak(_, [Tree2], NewTree1),
	make_mp2(A, [Tree1, NewTree1], NewTree).
make_mp3(A, [Tree1, Tree2], NewTree) :- % BとA⊃(B⊃C)からA⊃C 演繹定理で使用
	make_weak(_, [Tree1], NewTree1),
	make_mp2(A, [NewTree1, Tree2], NewTree).
make_id(A, NewTree) :-
	make_ax1(_=>p=>_, NewTree1), % pの箇所は何でもOK
	make_ax1(_, NewTree2),
	make_mp2(A, [NewTree1, NewTree2], NewTree).
make_pd1(A, [Tree1, Tree2], NewTree) :-
	make_ax3(_, NewTree1),
	make_mp(_, [Tree1, NewTree1], NewTree2),
	make_mp(A, [Tree2, NewTree2], NewTree).
make_pd2(A, [Tree1, Tree2], NewTree) :-
	make_weak(_, [Tree2], NewTree2),
	make_pd1(A, [Tree1, NewTree2], NewTree).
make_in(A, [Tree1, Tree2], NewTree) :-
	make_weak(_, [Tree1], NewTree1),
	make_pd2(A, [NewTree1, Tree2], NewTree).
make_in_subtheo(A=>B=>C, NewTree) :- % INから得る補助定理
	make_in(C, [
		proof_tree(A, hypo, [ ]),
		proof_tree(B, hypo, [ ])
	], NewTree1), % AとBからCを演繹(実際にはA=¬B)
	deduction_theorem(NewTree1, B, NewTree2), % ¬BからB⊃Cを演繹
	deduction_theorem(NewTree2, A, NewTree). % ¬B⊃B⊃Cを証明
make_dn(A, [Tree1], NewTree) :-
	make_weak(_, [Tree1], NewTree1),
	make_id(_, NewTree2),
	make_pd1(A, [NewTree1, NewTree2], NewTree).
make_dn_subtheo(A=>B, NewTree) :- % DNから得る補助定理
	make_dn(B, [
		proof_tree(A, hypo, [ ])
	], NewTree1), % AからBを演繹(実際にはA=¬¬B)
	deduction_theorem(NewTree1, A, NewTree). % ¬¬B⊃Bを証明

/* 演繹定理の適用。Γ,AからのBの演繹(様相論理の場合は正規演繹、以降でも同様)を
   第1引数に、Aを第2引数に与え、ΓからのA⊃Bの演繹を作って第3引数に入れる */
deduction_theorem(Tree, A, NewTree) :-
	reduce_hypo(Tree, A, NewTree1),
	(nonvar(NewTree1) ->
		NewTree = NewTree1 % A⊃Bの演繹を作った
	;
		make_weak(A=>_, [Tree], NewTree2), % A⊃Bの演繹を作った
		opt_tree(NewTree2, NewTree)
	).
/* deduction_theoremに似るが、仮定にAがない場合第3引数を未代入のままにする */
reduce_hypo(proof_tree(A, hypo, [ ]), B, NewTree) :- A == B, !,
	make_id(A=>_, NewTree).
reduce_hypo(proof_tree(_, _, [ ]), _, _) :- !.
reduce_hypo(proof_tree(_, mp, [Tree1, Tree2]), A, NewTree) :- !,
	reduce_hypo(Tree1, A, NewTree1), % NewTree1はA⊃Bの演繹または未代入
	reduce_hypo(Tree2, A, NewTree2), % NewTree2はA⊃B⊃Cの演繹または未代入
	(nonvar(NewTree1) ->
		(nonvar(NewTree2) ->
			make_mp2(_, [NewTree1, NewTree2], NewTree3)
		;
			make_mp1(_, [NewTree1, Tree2], NewTree3)
		), % A⊃Cの演繹を作った
		opt_tree(NewTree3, NewTree)
	;
		(nonvar(NewTree2) ->
			make_mp3(_, [Tree1, NewTree2], NewTree3),
			opt_tree(NewTree3, NewTree) % A⊃Cの演繹を作った
		;
			NewTree = _ % 仮定にAなし
		)
	).
reduce_hypo(proof_tree(_, nec, _), _, _). % 仮定はないはず
	
/* 証明木の最適化用。第1引数に与えた証明木の部分木で、第1引数と同じ論理式を
   既に証明できている部分があれば、そこを取り出し第2引数に入れる */
opt_tree(Tree, OptTree) :-
	Tree = proof_tree(A, _, _),
	conclu_subtree(A, Tree, _, OptTree, _).
/* 第2引数の木の部分木で、第1引数の論理式を結論に持つ最小のものを第4引数に、
   そのサイズを第5引数に入れる。なければそれらを未代入のままにし、木自体の
   サイズを第3引数へ */
conclu_subtree(A, Tree, Size, MinConcl, MinConclSize) :-
	Tree = proof_tree(B, _, L),
	list_conclu_subtree(A, L, S1, M1, MS1),
	(nonvar(M1) ->
		MinConcl = M1, MinConclSize = MS1
	;
		(A == B ->
			MinConcl = Tree, MinConclSize is S1 + 1
		;
			Size is S1 + 1
		)
	).
list_conclu_subtree(_, [ ], 0, _, _) :- !.
list_conclu_subtree(A, [Tree|L], SizeSum, MinConcl, MinConclSize) :-
	list_conclu_subtree(A, L, SizeSum1, MinConcl1, MinConclSize1),
	conclu_subtree(A, Tree, Size, MinConcl2, MinConclSize2),
	(nonvar(MinConcl1) ->
		(nonvar(MinConcl2), MinConclSize1 > MinConclSize2 ->
			MinConcl = MinConcl2, MinConclSize = MinConclSize2
		;
			MinConcl = MinConcl1, MinConclSize = MinConclSize1
		)
	;
		(nonvar(MinConcl2) ->
			MinConcl = MinConcl2, MinConclSize = MinConclSize2
		;
			SizeSum is SizeSum1 + Size
		)			
	).

/* Treeが(仮定なしの)証明木になっているかどうかのチェック */
check_tree(Tree) :- ground(Tree), check_tree_aux(Tree), !.
check_tree_aux(Tree) :- make_ax1(_, Tree).
check_tree_aux(Tree) :- make_ax2(_, Tree).
check_tree_aux(Tree) :- make_ax3(_, Tree).
check_tree_aux(Tree) :- make_k(_, Tree).
check_tree_aux(Tree) :- system_type(kd45), make_d(_, Tree).
check_tree_aux(Tree) :- system_type(kd45), make_4(_, Tree).
check_tree_aux(Tree) :- system_type(kd45), make_5(_, Tree).
check_tree_aux(Tree) :-
	make_mp(_, [Tree1, Tree2], Tree),
	check_tree_aux(Tree1),
	check_tree_aux(Tree2).
check_tree_aux(Tree) :-
	make_nec(_, [Tree1], Tree),
	check_tree_aux(Tree1).

/* 木の出力 */
output_tree(Tree) :- output_tree(Tree, 0, _).
output_tree(proof_tree(A, Rule, ChildrenTreeList), N, N1) :-
	output_treelist(ChildrenTreeList, N, N2, NL), N1 is N2+1,
	write(N1), write(' '), write_numslist(NL), write(' '), write(Rule),
	write(': '), write(A), nl.
output_treelist([ ], N, N, [ ]).
output_treelist([Tree|List], N, N1, [N2|NL]) :-
	output_tree(Tree, N, N2),
	output_treelist(List, N2, N1, NL).
write_numslist(L) :- write(L).

/* 定理(何でもよいので1つ選んで固定しておく)の証明木、あるいはその否定の作成 */
make_theorem(T, Tree) :- T = p=>p=>p, make_ax1(T, Tree).
not_theorem(~(p=>p=>p)).

/* 第1引数に与えたシーケントに対するHilbert流演繹を作成して第2引数に入れる
   第1引数はシーケントの→の左のリストと右のリストのペア
   シーケントの証明を作ってからHilbert流演繹に変換するのではなく
   シーケントの証明を作るためのWangのアルゴリズムを適用しつつ、その過程で
   Hilbert流演繹を直接作り出している
   3引数の場合の第3引数は(シーケント計算の)規則Nec, Nec45, NecD45を適用
   する直前のシーケントの→の左右それぞれから一番外の□を取り除いたリストのペア
   KD45の場合にこれが前回と同じになったらそこで規則の適用を止めることによって
   規則の適用の無限ループを防ぐためにある Kの場合は使っていない */
seq_deduction((Left, Right), NewTree) :-
	seq_deduction((Left, Right), NewTree, ([ ], [ ])).
seq_deduction((Left, Right), NewTree, _) :-
	member(A, Left), member(A, Right), !, % イニシャルシーケント
	not_theorem(NT),
	make_in(NT, [
		proof_tree(~A, hypo, [ ]), proof_tree(A, hypo, [ ])
	], NewTree). % ¬A,Aから¬Tの演繹を作った
seq_deduction((Left, Right), Tree, OrigLR) :-
	select(~A, Left, Left1), !, % →の左に¬Aあり
	% Γ,¬A,¬Δから¬Tの演繹を作る
	seq_deduction((Left1, [A|Right]), Tree, OrigLR).
seq_deduction((Left, Right), NewTree, OrigLR) :-
	select(~A, Right, Right1), !, % →の右に¬Aあり
	% Γ,A,¬Δから¬Tの演繹を作り、仮定のAを¬¬AからAの演繹に置き換え
	seq_deduction(([A|Left], Right1), Tree, OrigLR),
	make_dn(A, [proof_tree(~ ~A, hypo, [ ])], ReplTree), % ¬¬AからAの演繹
	replace_hypo(A, ReplTree, Tree, NewTree). % 仮定のAを¬¬Aに
seq_deduction((Left, Right), NewTree, OrigLR) :-
	select(A=>B, Right, Right1), !, % →の右にA⊃Bあり
	% Γ,A,¬Δ,¬Bから¬Tの演繹を作り、仮定のAと¬Bをそれぞれ
	% ¬(A⊃B)からA、¬(A⊃B)から¬Bの演繹に置き換え
	seq_deduction(([A|Left], [B|Right1]), Tree, OrigLR),
	make_weak(_, [proof_tree(~(A=>B), hypo, [ ])], ReplTree1),
	make_in_subtheo(_, ReplTree2),
	make_pd1(A, [ReplTree1, ReplTree2], ReplTree3), % ¬(A⊃B)からAの演繹
	make_weak(_, [proof_tree(~(A=>B), hypo, [ ])], ReplTree4),
	make_dn_subtheo(_, ReplTree5),
	make_ax1(_, ReplTree6),
	make_mp1(_, [ReplTree5, ReplTree6], ReplTree7),
	make_pd1(~B, [ReplTree4, ReplTree7], ReplTree8), %¬(A⊃B)から¬Bの演繹
	replace_hypo(A, ReplTree3, Tree, NewTree1), % 仮定のAを¬(A⊃B)に
	replace_hypo(~B, ReplTree8, NewTree1, NewTree). % 仮定の¬Bを¬(A⊃B)に
seq_deduction((Left, Right), NewTree, OrigLR) :-
	select(A=>B, Left, Left1), !, % →の左にA⊃Bあり
	% Γ,¬Δ,¬Aから¬Tの演繹を作り、Γ,¬Δから¬A⊃¬Tの演繹に作り替える
	seq_deduction((Left1, [A|Right]), Tree1, OrigLR),
	deduction_theorem(Tree1, ~A, ReplTree1),
	% Γ,B,¬Δから¬Tの演繹を作り、Γ,¬ΔからB⊃¬Tの演繹に作り替える
	seq_deduction(([B|Left1], Right), Tree2, OrigLR),
	deduction_theorem(Tree2, B, ReplTree2),
	% これらを用いてΓ,A⊃B,¬Δから¬Tの演繹を作る
	make_theorem(_, ReplTree3),
	make_pd2(_, [ReplTree1, ReplTree3], ReplTree4),
	make_mp(_, [ReplTree4, proof_tree(A=>B, hypo, [ ])], ReplTree5),
	make_mp(_, [ReplTree5, ReplTree2], NewTree). % Γ,A⊃B,¬Δから¬T演繹
seq_deduction((Left, Right), NewTree, OrigLR) :- system_type(k),
	nonnec(Left, Left0), sort(Left0, Left1),
	nonnec(Right, Right0), sort(Right0, Right1),
	member(B, Right1), % →の右に□Bあり ここでカットしない 
	% Γ,¬Bから¬Tの演繹を作り、Γから¬B⊃¬Tの演繹に作り替える
	seq_deduction((Left1, [B]), Tree, OrigLR), !, % OrigLRはKでは使わない
	deduction_theorem(Tree, ~B, NewTree1), % Γから¬B⊃¬Tの演繹
	make_theorem(_, NewTree2),
	make_pd2(_, [NewTree1, NewTree2], NewTree3), % ΓからBの演繹
	reverse(Left1, RevLeft1),
	repeat_deduction_theorem(RevLeft1, NewTree3, NewTree4), % Γ⊃Bの証明
	make_nec(_, [NewTree4], NewTree5),
	repeat_make_neck_mp(Left1, NewTree5, NewTree6), % □Γ⊃□Bの証明
	not_theorem(NT),
	make_in(NT, [proof_tree(~ #B, hypo, [ ]), NewTree6], NewTree).
seq_deduction((Left, Right), NewTree, (OrigL, [ ])) :- system_type(kd45),
	\+ member(#_, Right), !, % →の右に□Bなし
	nonnec1(Left, NLeft0), sort(NLeft0, NLeft1), nonnec(NLeft1, Left1),
	Left1 \= OrigL, merge_set(NLeft1, Left1, Left2), 
	% □Γ,Γから¬Tの演繹を作り、∧(□Γ∪Γ)⊃¬Tの証明に作り替える
	seq_deduction((Left2, [ ]), Tree, (Left1, [ ])),
	reverse(Left2, RevLeft2),
	repeat_deduction_theorem(RevLeft2, Tree, NewTree1),
	make_nec(_, [NewTree1], NewTree2), % □(∧(□Γ∪Γ)⊃¬T)の証明
	repeat_make_neck_mp(Left2, NewTree2, NewTree3),
	apply_45((Left1, [ ]), NewTree3, NewTree4), % □Γ⊃□¬Tの証明
	make_theorem(_, NewTree5), make_nec(_, [NewTree5], NewTree6),
	make_d(_, NewTree7), make_mp(_, [NewTree6, NewTree7], NewTree8),
	not_theorem(NT),
	make_in(NT, [NewTree8, NewTree4], NewTree). % □Γ⊃¬Tの証明
seq_deduction((Left, Right), NewTree, OrigLR) :- system_type(kd45),
	nonnec1(Left, NLeft0), sort(NLeft0, NLeft1), nonnec(NLeft1, Left1),
	nonnec1(Right, NRight0), sort(NRight0, NRight1),
	nonnec(NRight1, Right1),
	OrigLR \= (Left1, Right1), merge_set(NLeft1, Left1, Left2),
	member(B, Right1), % ここでカットしない
	% □Γ,Γ,¬□Σ,¬Bから¬Tの演繹を作り(ここでΣ=Δ∪{B})、
	% □Γ,Γ,¬□ΣからBの演繹に作り変える
	seq_deduction((Left2, [B|NRight1]), Tree, (Left1, Right1)), !,
	deduction_theorem(Tree, ~B, NewTree1), % □Γ,Γ,¬□Σから¬B⊃¬T
	make_theorem(_, NewTree2),
	make_pd2(_, [NewTree1, NewTree2], NewTree3), % □Γ,Γ,¬□ΣからB
	all_not(NRight1, NNRight1), reverse(NNRight1, RevNNRight1),
	repeat_deduction_theorem(RevNNRight1, NewTree3, NewTree4),
	reverse(Left2, RevLeft2),
	repeat_deduction_theorem(RevLeft2, NewTree4, NewTree5),
	make_nec(_, [NewTree5], NewTree6), % □(∧(□Γ∪Γ∪¬□Σ)⊃B)
	repeat_make_neck_mp(Left2, NewTree6, NewTree7),
	repeat_make_neck_mp(NNRight1, NewTree7, NewTree8),
	apply_45((Left1, Right1), NewTree8, NewTree9), % □Γ,¬□Σから□B
	make_in(_, [proof_tree(~ #B, hypo, [ ]), NewTree9], NewTree).
all_not([ ], [ ]).
all_not([A|L], [~A|L1]) :- all_not(L, L1).
nonnec([ ], [ ]).
nonnec([#A|L], [A|L1]) :- !, nonnec(L, L1).
nonnec([_|L], L1) :- nonnec(L, L1).
nonnec1([ ], [ ]).
nonnec1([#A|L], [#A|L1]) :- !, nonnec1(L, L1).
nonnec1([_|L], L1) :- nonnec1(L, L1).
repeat_deduction_theorem([ ], Tree, Tree).
repeat_deduction_theorem([A|L], Tree, NewTree) :-
	deduction_theorem(Tree, A, NewTree1),
	repeat_deduction_theorem(L, NewTree1, NewTree).
repeat_make_neck_mp([ ], Tree, Tree).
repeat_make_neck_mp([A|L], Tree, NewTree) :-
	make_k(_, NewTree1),
	make_mp(_, [Tree, NewTree1], NewTree2),
	make_mp(_, [proof_tree(#A, hypo, [ ]), NewTree2], NewTree3),
	repeat_make_neck_mp(L, NewTree3, NewTree).
apply_45((L, _), Tree, NewTree) :- Tree = proof_tree(# #A, hypo, [ ]), !,
	(member(A, L) ->
		make_4(_, NewTree1),
		make_mp(_, [proof_tree(#A, hypo, [ ]), NewTree1], NewTree)
	;
		NewTree = Tree
	).
apply_45((_, R), Tree, NewTree) :- Tree = proof_tree(# ~ #A, hypo, [ ]), !,
	(member(A, R) ->
		make_5(_, NewTree1),
		make_mp(_, [proof_tree(~ #A, hypo, [ ]), NewTree1], NewTree)
	;
		NewTree = Tree
	).
apply_45(LR, proof_tree(A, Type, List), proof_tree(A, Type, NewList)) :-
	apply_45_list(LR, List, NewList).
apply_45_list(_, [ ], [ ]) :- !.
apply_45_list(LR, [Tree|List], [Tree1|List1]) :-
	apply_45(LR, Tree, Tree1),
	apply_45_list(LR, List, List1).

/* 第3引数に与えた演繹の木のうち、第1引数の論理式が葉になっているところに
   ついて、それを第2引数の木で置き換えたものを作る */
replace_hypo(A, ReplTree, proof_tree(A, hypo, [ ]), ReplTree) :- !.
replace_hypo(_, _, Tree, Tree) :- Tree = proof_tree(_, _, [ ]), !.
replace_hypo(A, ReplTree, Tree, NewTree) :-
	Tree = proof_tree(B, Type, List),
	NewTree = proof_tree(B, Type, NewList),
	list_replace_hypo(A, ReplTree, List, NewList).
list_replace_hypo(_, _, [ ], [ ]) :- !.
list_replace_hypo(A, ReplTree, [Tree|Rest], [NewTree|NewRest]) :-
	replace_hypo(A, ReplTree, Tree, NewTree),
	list_replace_hypo(A, ReplTree, Rest, NewRest).

/* 論理式Aの証明を作る Aには∧∨◇が入っていてもよい */
deduction(A, Tree) :-
	expand_abbrev(A, B), % A中の略記を展開してBを得る
	seq_deduction(([ ], [B]), Tree1), % ¬Bから¬Tへの演繹を作る
	deduction_theorem(Tree1, ~B, Tree2), % ¬B⊃¬Tの証明を作る
	make_theorem(_, Tree3),
	make_pd2(_, [Tree2, Tree3], Tree4), % これでBの証明が完成
	opt_tree(Tree4, Tree),
	(check_tree(Tree) -> % 念のためTreeの正当性を検査
		true
	;
		write('A bug occurred while proving '), write(A), nl, fail
	).

/* 論理式Aの証明を作って出力 */
prove(A) :-
	deduction(A, Tree),
	output_tree(Tree).

set_system_type(SysType) :-
	member(SysType, [k, kd45]), !,
	abolish(system_type/1),
	assert(system_type(SysType)).
set_system_type_if_unset(SysType) :-
	(clause(system_type(_), _) -> true; set_system_type(SysType)).
:- dynamic(system_type/1).
:- set_system_type_if_unset(k).


